# $Id: PKGBUILD 152793 2012-03-09 16:30:31Z andyrtr $
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=libxtst
pkgver=1.2.1
pkgrel=1
pkgdesc="X11 Testing -- Resource extension library"
arch=('i686' 'x86_64')
url="http://xorg.freedesktop.org/"
license=('custom')
depends=('libxext' 'libxi' 'recordproto' 'inputproto')
makedepends=('xorg-util-macros')
options=('!libtool')
source=("${url}/releases/individual/lib/libXtst-${pkgver}.tar.bz2")
sha1sums=('44e19899f9938d3db5546e51c338d6f86d88cfe2')

build() {
  cd "${srcdir}/libXtst-${pkgver}"
  ./configure --prefix=/usr --disable-static
  make
}

package() {
  cd "${srcdir}/libXtst-${pkgver}"
  make DESTDIR="${pkgdir}" install

  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
